
Scenario:

While writing text to an external file in a DATA step, you want to generate diagnostic messages to the SAS log
when suspicious data is encountered.

The destination of text written by the PUT statement is determined by the previously executed FILE statement.

The old way: Switch Between the External File and the SAS Log

The original approach is to execute a FILE LOG statement prior to each diagnostic
message, and then re-execute the FILE statement for the output file after the diagnostic
message is complete.

data _null_;
  file "c:\temp\Dinosaur\PutLogFile1.csv" dsd;
  if _n_=1 then put "Name,Age,Height";
  set sashelp.class;
  if age lt 12 then do;
    file log;
    put "Student age is less than 12 ... Please verify for " name= age=;
    file "c:\temp\Dinosaur\PutLogFile1.csv";
  end;
  put Name Age Height;
run;

The new way: Write to both the External File and the SAS Log
(available in SAS 9)

An alternate approach is to use the PUTLOG statement for the diagostic messages.

data _null_;
  file "c:\temp\Dinosaur\PutLogFile2.csv" dsd;
  if _n_=1 then put "Name,Age,Height";
  set sashelp.class;
  if age lt 12 then
    putlog "Student age is less than 12 ... Please verify for " name= age=;
  put Name Age Height;
run;

Advantages of the alternate approach:

• The PUTLOG statement writes only to the SAS log, eliminating any potential interference with the
external output file.

• The only FILE statement required corresponds to the external output file. The original approach may
require the main FILE statement to be executed multiple times.

Disadvantages of the alternate approach:

• The PUTLOG statement writes only to the SAS log, so you cannot redirect diagnostic messages to a
secondary external file. You must specify multiple FILE statements, one for the main external file and
one for the diagnostics.

Additional documentation for this technique can be found in  SAS® 9.2 Language Reference: Dictionary. Cary, NC: SAS Institute
Inc.

Visit http://support.sas.com/documentation/onlinedoc/sas9doc.html for SAS 9 documentation.

Go to  http://www.repole.com/dinosaur/putlog.html  for the sample code and output for this topic.

These techniques are mentioned in other SAS references and publications:
SAS Sample 24739(http://support.sas.com/kb/24/739.html)
In the Know: SAS Tips and Techniques From Around The Globe, Second Edition; Phil Mason, 2007 .
The DATA step in Version 9: What's New?; Jason Secosky,
http://support.sas.com/rnd/papers/sugi27/dsv9-sugi.pdf .


